

# NAG Fortran Library Routine Document

## F08QGF (STRSEN/DTRSEN)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F08QGF (STRSEN/DTRSEN) reorders the Schur factorization of a real general matrix so that a selected cluster of eigenvalues appears in the leading elements or blocks on the diagonal of the Schur form. The routine also optionally computes the reciprocal condition numbers of the cluster of eigenvalues and/or the invariant subspace.

### 2 Specification

```

SUBROUTINE F08QGF (JOB, COMPQ, SELECT, N, T, LDT, Q, LDQ, WR, WI, M, S,
1              SEP, WORK, LWORK, IWORK, LIWORK, INFO)
ENTRY          strsen (JOB, COMPQ, SELECT, N, T, LDT, Q, LDQ, WR, WI, M, S,
1              SEP, WORK, LWORK, IWORK, LIWORK, INFO)
INTEGER        N, LDT, LDQ, M, LWORK, IWORK(*), LIWORK, INFO
real          T(LDT,*), Q(LDQ,*), WR(*), WI(*), S, SEP, WORK(*)
LOGICAL        SELECT(*)
CHARACTER*1    JOB, COMPQ

```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

This routine reorders the Schur factorization of a real general matrix  $A = QTQ^T$ , so that a selected cluster of eigenvalues appears in the leading diagonal elements or blocks of the Schur form.

The reordered Schur form  $\tilde{T}$  is computed by an orthogonal similarity transformation:  $\tilde{T} = Z^T T Z$ . Optionally the updated matrix  $\tilde{Q}$  of Schur vectors is computed as  $\tilde{Q} = QZ$ , giving  $A = \tilde{Q}\tilde{T}\tilde{Q}^T$ .

Let  $\tilde{T} = \begin{pmatrix} T_{11} & T_{12} \\ 0 & T_{22} \end{pmatrix}$ , where the selected eigenvalues are precisely the eigenvalues of the leading  $m$  by  $m$  submatrix  $T_{11}$ . Let  $\tilde{Q}$  be correspondingly partitioned as  $(Q_1 \ Q_2)$  where  $Q_1$  consists of the first  $m$  columns of  $\tilde{Q}$ . Then  $AQ_1 = Q_1 T_{11}$ , and so the  $m$  columns of  $Q_1$  form an orthonormal basis for the invariant subspace corresponding to the selected cluster of eigenvalues.

Optionally the routine also computes estimates of the reciprocal condition numbers of the average of the cluster of eigenvalues and of the invariant subspace.

### 4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

1: JOB – CHARACTER\*1 *Input*

*On entry:* indicates whether condition numbers are required for the cluster of eigenvalues and/or the invariant subspace, as follows:

if JOB = 'N', no condition numbers are required;

if JOB = 'E', only the condition number for the cluster of eigenvalues is computed;

if JOB = 'V', only the condition number for the invariant subspace is computed;  
 if JOB = 'B', condition numbers for both the cluster of eigenvalues and the invariant subspace are computed.

*Constraint:* JOB = 'N', 'E', 'V' or 'B'.

2: COMPQ – CHARACTER\*1 *Input*

*On entry:* indicates whether the matrix  $Q$  of Schur vectors is to be updated, as follows:

if COMPQ = 'V', the matrix  $Q$  of Schur vectors is updated;

if COMPQ = 'N', no Schur vectors are updated.

*Constraint:* COMPQ = 'V' or 'N'.

3: SELECT(\*) – LOGICAL array *Input*

**Note:** the dimension of the array SELECT must be at least  $\max(1, N)$ .

*On entry:* the eigenvalues in the selected cluster. To select a real eigenvalue  $\lambda_j$ , SELECT( $j$ ) must be set .TRUE.. To select a complex conjugate pair of eigenvalues  $\lambda_j$  and  $\lambda_{j+1}$  (corresponding to a 2 by 2 diagonal block), SELECT( $j$ ) and/or SELECT( $j+1$ ) must be set to .TRUE.. A complex conjugate pair of eigenvalues **must** be either both included in the cluster or both excluded. See also Section 8.

4: N – INTEGER *Input*

*On entry:*  $n$ , the order of the matrix  $T$ .

*Constraint:*  $N \geq 0$ .

5: T(LDT,\*) – *real* array *Input/Output*

**Note:** the second dimension of the array T must be at least  $\max(1, N)$ .

*On entry:* the  $n$  by  $n$  upper quasi-triangular matrix  $T$  in canonical Schur form, as returned by F08PEF (SHSEQR/DHSEQR). See also Section 8.

*On exit:*  $T$  is overwritten by the updated matrix  $\tilde{T}$ .

6: LDT – INTEGER *Input*

*On entry:* the first dimension of the array T as declared in the (sub)program from which F08QGF (STRSEN/DTRSEN) is called.

*Constraint:*  $LDT \geq \max(1, N)$ .

7: Q(LDQ,\*) – *real* array *Input/Output*

**Note:** the second dimension of the array Q must be at least  $\max(1, N)$  if COMPQ = 'V', and at least 1 if COMPQ = 'N'.

*On entry:* if COMPQ = 'V', Q must contain the  $n$  by  $n$  orthogonal matrix  $Q$  of Schur vectors, as returned by F08PEF (SHSEQR/DHSEQR).

*On exit:* if COMPQ = 'V', Q contains the updated matrix of Schur vectors; the first  $m$  columns of Q form an orthonormal basis for the specified invariant subspace.

Q is not referenced if COMPQ = 'N'.

- 8: LDQ – INTEGER *Input*
- On entry:* the first dimension of the array Q as declared in the (sub)program from which F08QGF (STRSEN/DTRSEN) is called.
- Constraints:*
- $$\begin{aligned} \text{LDQ} &\geq \max(1, N) \text{ if } \text{COMPQ} = \text{'V'}, \\ \text{LDQ} &\geq 1 \text{ if } \text{COMPQ} = \text{'N'}. \end{aligned}$$
- 9: WR(\*) – *real* array *Output*
- 10: WI(\*) – *real* array *Output*
- Note:** the dimensions of the arrays WR and WI must each be at least  $\max(1, N)$ .
- On exit:* the real and imaginary parts, respectively, of the reordered eigenvalues of  $\tilde{T}$ . The eigenvalues are stored in the same order as on the diagonal of  $\tilde{T}$ ; see Section 8 for details. Note that if a complex eigenvalue is sufficiently ill-conditioned, then its value may differ significantly from its value before reordering.
- 11: M – INTEGER *Output*
- On exit:*  $m$ , the dimension of the specified invariant subspace. The value of  $m$  is obtained by counting 1 for each selected real eigenvalue and 2 for each selected complex conjugate pair of eigenvalues (see SELECT);  $0 \leq m \leq n$ .
- 12: S – *real* *Output*
- On exit:* if JOB = 'E' or 'B', S is a lower bound on the reciprocal condition number of the average of the selected cluster of eigenvalues. If M = 0 or N, S = 1; if INFO = 1 (see Section 6), S is set to zero.
- S is not referenced if JOB = 'N' or 'V'.
- 13: SEP – *real* *Output*
- On exit:* if JOB = 'V' or 'B', SEP is the estimated reciprocal condition number of the specified invariant subspace. If M = 0 or N,  $\text{SEP} = \|T\|$ ; if INFO = 1 (see Section 6), SEP is set to zero.
- SEP is not referenced if JOB = 'N' or 'E'.
- 14: WORK(\*) – *real* array *Workspace*
- Note:** the dimension of the array WORK must be at least  $\max(1, \text{LWORK})$ .
- On exit:* if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimum performance.
- 15: LWORK – INTEGER *Input*
- On entry:* the dimension of the array WORK as declared in the (sub)program from which F08QGF (STRSEN/DTRSEN) is called, unless LWORK = -1, in which case a workspace query is assumed and the routine only calculates the minimum dimension of WORK.
- Constraints:*
- $$\begin{aligned} \text{if } \text{JOB} = \text{'N'}, & \text{ then } \text{LWORK} \geq \max(1, N) \text{ or } \text{LWORK} = -1, \\ \text{if } \text{JOB} = \text{'E'}, & \text{ then } \text{LWORK} \geq \max(1, m \times (N - m)) \text{ or } \text{LWORK} = -1, \\ \text{if } \text{JOB} = \text{'V' or 'B'}, & \text{ then } \text{LWORK} \geq \max(1, 2m \times (N - m)) \text{ or } \text{LWORK} = -1. \end{aligned}$$
- The actual amount of workspace required cannot exceed  $N^2/4$  if JOB = 'E' or  $N^2/2$  if JOB = 'V' or 'B'.

16: IWORK(\*) – INTEGER array *Workspace*

**Note:** the dimension of the array IWORK must be at least  $\max(1, \text{LIWORK})$ .

*On exit:* if INFO = 0, IWORK(1) contains the required minimal size of LIWORK.

17: LIWORK – INTEGER *Input*

*On entry:* the dimension of the array IWORK as declared in the (sub)program from which F08QGF (STRSEN/DTRSEN) is called, unless LIWORK = -1, in which case a workspace query is assumed and the routine only calculates the minimum dimension of IWORK.

*Constraints:*

if JOB = 'N' or 'E', then  $\text{LIWORK} \geq 1$  or  $\text{LIWORK} = -1$ ,

if JOB = 'V' or 'B', then  $\text{LIWORK} \geq \max(1, m \times (N - m))$  or  $\text{LIWORK} = -1$ .

The actual amount of workspace required cannot exceed  $N^2/2$  if JOB = 'V' or 'B'.

18: INFO – INTEGER *Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If  $\text{INFO} = -i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO = 1

The reordering of  $T$  failed because a selected eigenvalue was too close to an eigenvalue which was not selected; this error exit can only occur if at least one of the eigenvalues involved was complex. The problem is too ill-conditioned: consider modifying the selection of eigenvalues so that eigenvalues which are very close together are either all included in the cluster or all excluded. On exit,  $T$  may have been partially reordered, but WR, WI and  $Q$  (if requested) are updated consistently with  $T$ ; S and SEP (if requested) are both set to zero.

## 7 Accuracy

The computed matrix  $\tilde{T}$  is similar to a matrix  $T + E$ , where

$$\|E\|_2 = O(\epsilon)\|T\|_2,$$

and  $\epsilon$  is the *machine precision*.

S cannot underestimate the true reciprocal condition number by more than a factor of  $\sqrt{\min(m, n - m)}$ . SEP may differ from the true value by  $\sqrt{m(n - m)}$ . The angle between the computed invariant subspace and the true subspace is  $\frac{O(\epsilon)\|A\|_2}{sep}$ .

Note that if a 2 by 2 diagonal block is involved in the re-ordering, its off-diagonal elements are in general changed; the diagonal elements and the eigenvalues of the block are unchanged unless the block is sufficiently ill-conditioned, in which case they may be noticeably altered. It is possible for a 2 by 2 block to break into two 1 by 1 blocks, that is, for a pair of complex eigenvalues to become purely real. The values of real eigenvalues however are never changed by the re-ordering.

## 8 Further Comments

The input matrix  $T$  must be in canonical Schur form, as is the output matrix  $\tilde{T}$ . This has the following structure.

If all the computed eigenvalues are real,  $\tilde{T}$  is upper triangular, and the diagonal elements of  $\tilde{T}$  are the eigenvalues;  $\text{WR}(i) = \tilde{t}_{ii}$  for  $i = 1, 2, \dots, n$  and  $\text{WI}(i) = 0.0$ .

If some of the computed eigenvalues form complex conjugate pairs, then  $\tilde{T}$  has 2 by 2 diagonal blocks. Each diagonal block has the form

$$\begin{pmatrix} \tilde{t}_{ii} & \tilde{t}_{i,i+1} \\ \tilde{t}_{i+1,i} & \tilde{t}_{i+1,i+1} \end{pmatrix} = \begin{pmatrix} \alpha & \beta \\ \gamma & \alpha \end{pmatrix}$$

where  $\beta\gamma < 0$ . The corresponding eigenvalues are  $\alpha \pm \sqrt{\beta\gamma}$ ;  $\text{WR}(i) = \text{WR}(i+1) = \alpha$ ;  $\text{WI}(i) = +\sqrt{|\beta\gamma|}$ ;  $\text{WI}(i+1) = -\text{WI}(i)$ .

The complex analogue of this routine is F08QUF (CTRSEN/ZTRSEN).

## 9 Example

To reorder the Schur factorization of the matrix  $A = QTQ^T$  such that the two real eigenvalues appear as the leading elements on the diagonal of the reordered matrix  $\tilde{T}$ , where

$$T = \begin{pmatrix} 0.7995 & -0.1144 & 0.0060 & 0.0336 \\ 0.0000 & -0.0994 & 0.2478 & 0.3474 \\ 0.0000 & -0.6483 & -0.0994 & 0.2026 \\ 0.0000 & 0.0000 & 0.0000 & -0.1007 \end{pmatrix}$$

and

$$Q = \begin{pmatrix} 0.6551 & 0.1037 & 0.3450 & 0.6641 \\ 0.5236 & -0.5807 & -0.6141 & -0.1068 \\ -0.5362 & -0.3073 & -0.2935 & 0.7293 \\ 0.0956 & 0.7467 & -0.6463 & 0.1249 \end{pmatrix}.$$

The original matrix  $A$  is given in F08NFF (SORGHR/DORGHR).

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F08QGF Example Program Text
*      Mark 16 Release. NAG Copyright 1992.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5, NOUT=6)
      INTEGER          NMAX, LDT, LDQ, LWORK, LIWORK
      PARAMETER        (NMAX=8, LDT=NMAX, LDQ=NMAX, LWORK=NMAX*NMAX/2,
+                     LIWORK=NMAX*NMAX/2)
*      .. Local Scalars ..
      real            S, SEP
      INTEGER          I, IFAIL, INFO, J, M, N
*      .. Local Arrays ..
      real            Q(LDQ, NMAX), T(LDT, NMAX), WI(NMAX), WORK(LWORK),
+                     WR(NMAX)
      INTEGER          IWORK(LIWORK)
      LOGICAL          SELECT(NMAX)
*      .. External Subroutines ..
      EXTERNAL         strsen, X04CAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F08QGF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
```

```

      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*       Read T and Q from data file
*
      READ (NIN,*) ((T(I,J),J=1,N),I=1,N)
      READ (NIN,*) ((Q(I,J),J=1,N),I=1,N)
*
      READ (NIN,*) (SELECT(I),I=1,N)
*
*       Reorder the Schur factorization
*
      CALL strsen('Both','Vectors',SELECT,N,T,LDT,Q,LDQ,WR,WI,M,S,
+              SEP,WORK,LWORK,IWORK,LIWORK,INFO)
*
      WRITE (NOUT,*)
      IFAIL = 0
*
      CALL X04CAF('General',' ',N,N,T,LDT,'Reordered Schur form',
+              IFAIL)
*
      WRITE (NOUT,*)
      IFAIL = 0
*
      CALL X04CAF('General',' ',N,M,Q,LDQ,
+              'Basis of invariant subspace',IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,99999) 'Condition number estimate',
+      ' of the selected cluster of eigenvalues = ', 1.0e0/S
      WRITE (NOUT,*)
      WRITE (NOUT,99999) 'Condition number estimate of the spec',
+      'ified invariant subspace = ', 1.0e0/SEP
      END IF
      STOP
*
99999 FORMAT (1X,A,A,e10.2)
      END

```

## 9.2 Program Data

F08QGF Example Program Data

```

4
0.7995  -0.1144  0.0060  0.0336      :Value of N
0.0000  -0.0994  0.2478  0.3474
0.0000  -0.6483  -0.0994  0.2026
0.0000   0.0000  0.0000  -0.1007      :End of matrix T
0.6551   0.1037  0.3450  0.6641
0.5236  -0.5807  -0.6141  -0.1068
-0.5362  -0.3073  -0.2935  0.7293
0.0956   0.7467  -0.6463  0.1249      :End of matrix Q
T   F   F   T      :End of SELECT

```

### 9.3 Program Results

F08QGF Example Program Results

Reordered Schur form

	1	2	3	4
1	0.7995	-0.0059	0.0751	-0.0927
2	-0.0000	-0.1007	-0.3936	0.3569
3	0.0000	0.0000	-0.0994	0.5128
4	0.0000	0.0000	-0.3133	-0.0994

Basis of invariant subspace

	1	2
1	0.6551	0.1211
2	0.5236	0.3286
3	-0.5362	0.5974
4	0.0956	0.7215

Condition number estimate of the selected cluster of eigenvalues = 0.18E+01

Condition number estimate of the specified invariant subspace = 0.32E+01

---